
UNIVERSIDAD NACIONAL MAYOR DE SAN MARCOS

Universidad del Perú, Decana de América

Facultad de Ingeniería Electrónica y Eléctrica
Escuela Profesional de Ingeniería Electrónica

GUÍA DE LABORATORIO N.º 1

Lógica CMOS Estática

*Caracterización, dimensionamiento y potencia
con modelos predictivos PTM en LTspice*

Curso: Microelectrónica
Unidad: Modelamiento de circuitos
Docente: MsC. Luz Adanaqué Infante
Integrantes: Leonardo Sait Yactayo Tolentino (23190214)
Marco Antonio Arcila Santander (23190140)
Modalidad: Parejas

1. Objetivos

Al terminar la sesión el estudiante será capaz de:

1. Configurar un entorno de simulación con tarjetas de modelo predictivas (PTM) para distintos nodos tecnológicos.
2. Extraer la curva de transferencia (VTC) de un inversor CMOS estático y determinar V_M , V_{IL} , V_{IH} y los márgenes de ruido.
3. Medir retardos de propagación, tiempos de transición y su dependencia con la carga y el nodo tecnológico.
4. Sintetizar compuertas complejas (NAND, NOR, AOI) mediante redes pull-up / pull-down duales y verificarlas funcionalmente.
5. Cuantificar las tres componentes de potencia (dinámica, cortocircuito y fuga) y contrastar procesos HP frente a LP.

2. Fundamento resumido y Pre-lab

La lógica CMOS estática implementa una función lógica $Y = \overline{F(A, B, \dots)}$ mediante dos redes complementarias: una red *pull-down* (PDN) de transistores NMOS que conduce hacia tierra cuando $F = 1$, y una red *pull-up* (PUN) de transistores PMOS, grafo dual de la anterior, que conmuta hacia V_{DD} cuando $F = 0$. En régimen estático solo una de las dos redes conduce plenamente, de modo que en el modelo ideal no existe camino DC entre V_{DD} y tierra ($I_{DC} \approx 0$).

Deducción analítica del punto de conmutación V_M (Pre-lab)

El punto de conmutación o umbral lógico V_M se define como el punto de la característica de transferencia de tensión (VTC) donde el voltaje de salida iguala al voltaje de entrada:

$$V_{in} = V_{out} = V_M \quad (1)$$

1. Hipótesis operativas y saturación por velocidad:

En el inversor nanométrico (Figura 1) se adopta geometría de canal mínimo $L_n = L_p = L_{\min} = 45$ nm y sustratos conectados a sus fuentes ($V_{SBn} = V_{SBp} = 0$, anulando el efecto de cuerpo, $\Delta V_T = 0$). Bajo fuerte campo longitudinal ($\mathcal{E} \gg \mathcal{E}_{sat}$), la velocidad de deriva satura en $v_{sat} = \mu \mathcal{E}_{sat} = \mu \frac{V_{DSAT}}{L}$, permitiendo linealizar la corriente de drenador:

$$I_D \approx k V_{DSAT} (V_{GS} - V_T) = W C_{ox} v_{sat} (V_{GS} - V_T) \quad (2)$$

donde la equivalencia $k V_{DSAT} = C_{ox} W v_{sat}$ es válida al ser $L_n = L_p = L_{\min}$.

2. Equilibrio estático de corrientes:

Al no existir corriente de compuerta estacionaria hacia la carga ($I_{out} = 0$), las corrientes de drenador se igualan en el punto de umbral ($I_{Dn} = I_{Dp}$):

$$I_{Dn} = k_n V_{DSATn} (V_{GSn} - V_{Tn}) = k_n V_{DSATn} (V_M - V_{Tn}) \quad (3)$$

$$I_{Dp} = k_p V_{DSATp} (V_{SGp} - |V_{Tp}|) = k_p V_{DSATp} (V_{DD} - V_M + V_{Tp}) \quad (4)$$

donde $-|V_{Tp}| = +V_{Tp}$ ($V_{Tp} < 0$). Definiendo la razón de transconductancia de saturación r :

$$r \equiv \frac{k_p V_{DSATp}}{k_n V_{DSATn}} = \frac{C_{ox} W_p v_{sat,p}}{C_{ox} W_n v_{sat,n}} = \frac{W_p}{W_n} \cdot \frac{v_{sat,p}}{v_{sat,n}} = \beta \cdot \frac{v_{sat,p}}{v_{sat,n}} \quad (5)$$

con $\beta \equiv W_p/W_n$. Sustituyendo r en $I_{Dn} = I_{Dp}$ y despejando analíticamente:

$$V_M = \frac{V_{Tn} + r (V_{DD} + V_{Tp})}{1 + r}, \quad r = \beta \cdot \frac{v_{sat,p}}{v_{sat,n}} \quad (6)$$

3. Condición de dimensionamiento para centrado simétrico ($V_M = V_{DD}/2$):

Para maximizar y equilibrar simétricamente los márgenes de ruido ($NM_L = NM_H$), se impone $V_M = V_{DD}/2$:

$$\beta_{\text{sim}} = \left(\frac{\frac{V_{DD}}{2} - V_{Tn}}{\frac{V_{DD}}{2} - |V_{Tp}|} \right) \cdot \frac{v_{\text{sat},n}}{v_{\text{sat},p}} \quad (7)$$

Bajo umbrales simétricos ($V_{Tn} \approx |V_{Tp}|$), la razón exige $\beta_{\text{sim}} \approx v_{\text{sat},n}/v_{\text{sat},p} \approx 1.13$, convergiendo en la simulación nanométrica a $\beta \approx 2.50$ debido a los efectos combinados de canal corto. La hipótesis de saturación simultánea en $V_M \approx 0.50$ V queda plenamente verificada ($V_{DSn} \geq V_{DSATn} \approx 0.12$ V y $V_{SDp} \geq V_{DSATp} \approx 0.15$ V), asegurando la consistencia física del modelo.




Figura 1: Esquema circuitual del inversor CMOS (45 nm HP) en condición de conmutación ($V_{in} = V_{out} = V_M$, $I_{Dn} = I_{Dp}$).

Dinámica de retardo y componentes de potencia

Retardo de propagación y descomposición capacitiva. El retardo medio de propagación t_p se define como el promedio entre los tiempos de respuesta de subida y bajada:

$$t_p = \frac{1}{2} (t_{pHL} + t_{pLH}), \quad t_{pHL} \approx 0.69 R_{eq,n} C_L, \quad t_{pLH} \approx 0.69 R_{eq,p} C_L \quad (8)$$

donde $R_{eq,n} \propto \frac{1}{W_n}$ y $R_{eq,p} \propto \frac{1}{\beta W_n}$. En un circuito integrado, la capacitancia total de carga C_L descompone en componentes intrínsecas y externas:

$$C_L = C_{ext} + C_{int} = C_{ext} + \underbrace{(C_{db,n} + C_{db,p})}_{\text{difusión}} + \underbrace{(C_{gd,n} + C_{gd,p})}_{\text{solapamiento}} + C_{wire} \quad (9)$$

Esta descomposición justifica que en $t_p(C_L) = m C_L + t_0$, la ordenada $t_0 \approx 0.69 R_{eq} C_{int}$ represente el retardo intrínseco descargado.

Componentes fundamentales de potencia disipada. La potencia media disipada por compuerta CMOS en conmutación periódica a frecuencia f consta de tres términos físicos bien diferenciados:

$$P_{tot} = \underbrace{\alpha C_L V_{DD}^2 f}_{\text{dinámica de conmutación}} + \underbrace{I_{cc} V_{DD}}_{\text{cortocircuito en transición}} + \underbrace{I_{leak} V_{DD}}_{\text{fuga estática}} \quad (10)$$

3. Entorno y Modelos Tecnológicos

Las simulaciones se ejecutaron en LTspice 24 mediante modelos predictivos PTM (BSIM4) para 90 nm bulk, 45 nm (HP/LP) y 22 nm HP. Se adoptó un ancho unitario de referencia $W_n = 4L_{\text{mín}}$ y se renombraron los modelos (45nm_HP_ren.lib y 45nm_LP_ren.lib) para evitar colisiones de parámetros en directivas .include.

4. Actividad 1 — Inversor CMOS: VTC y márgenes de ruido

Procedimiento

1. Arme el circuito del Anexo A.1 con la tarjeta de 45 nm HP, $V_{DD} = 1.0$ V, $L = 45$ nm, $W_n = 180$ nm.
2. Ejecute el barrido DC con `.step param beta 1 3 0.5` y grafique la familia de curvas $V_{out}(V_{in})$ (Figura 2).
3. Registre V_M para cada valor de β mediante la directiva `.meas`. Los resultados se presentan en la Tabla 1.
4. Para el caso de β que hace $V_M \approx V_{DD}/2$: grafique la derivada $d(V_{out})/d(V_{in})$ en el visor de formas de onda y ubique con cursores los dos puntos donde la pendiente vale -1 . Esos son V_{IL} y V_{IH} , capturados directamente en LTspice (Figura 3).
5. Calcule $NM_H = V_{OH} - V_{IH}$ y $NM_L = V_{IL} - V_{OL}$.

Tabla 1: Resultados 1.1 — Barrido de la razón β .

β	V_M (V)	V_{IL} (V)	V_{IH} (V)	NM_L (V)	NM_H (V)	Gan. máx.
1.0	0.4632	0.3460	0.5560	0.3460	0.4440	6.18
1.5	0.4791	0.3680	0.5760	0.3680	0.4240	6.13
2.0	0.4903	0.3820	0.5920	0.3820	0.4080	6.10
2.5	0.4989	0.3920	0.6040	0.3920	0.3960	6.08
3.0	0.5058	0.4020	0.6140	0.4020	0.3860	6.06

Tabla 2: Resultados 1.2 — Parámetros extraídos de 45nm_HP.pm.

Parámetro	NMOS	PMOS	Razón n/p
u0 (movilidad)	0.0540 m ² /V·s	0.0200 m ² /V·s	2.70
vsat	1.70×10^5 m/s	1.50×10^5 m/s	1.13
vth0	0.4689 V	-0.4916 V	0.95




Figura 2: Curvas de transferencia de tensión (VTC) y derivada $d(V_{out})/d(V_{in})$ para el inversor de 45 nm HP ante el barrido de $\beta \in [1.0, 3.0]$.




Figura 3: Medición directa en LTspice de los voltajes críticos V_{IL} y V_{IH} mediante cursores para cada β simulado.

Preguntas de la Actividad 1

P1. Con la razón de u_0 obtenida, la regla clásica predice cierto β para centrar la VTC. ¿Coincide con el β que usted midió? Explique la discrepancia en términos de saturación de velocidad, apoyándose en la ecuación (6).

Rpta: La regla clásica predice $\beta \approx \mu_n/\mu_p = 2.70$, pero la simulación ubica el centrado en $\beta \approx 2.50$. Esta discrepancia se fundamenta en las ecuaciones (6) y (7): en 45 nm domina la saturación de velocidad, por lo que la razón efectiva r depende de $v_{sat,p}/v_{sat,n} = 1.13^{-1} \approx 0.88$, convergiendo a un factor β sensiblemente menor al predicho por la movilidad.

P2. ¿Por qué V_{OH} y V_{OL} son exactamente V_{DD} y 0 V en lógica CMOS estática, y no lo son en lógica pseudo-NMOS?

Rpta: En reposo estático, una red conduce plenamente mientras la complementaria está en corte riguroso ($I_{DC} \approx 0$), alcanzando niveles riel a riel sin caída resistiva ($V_{OH} = V_{DD}$, $V_{OL} = 0$ V). En pseudo-NMOS el PMOS conduce continuamente; al activarse la PDN se forma un divisor de tensión resistivo que degrada $V_{OL} > 0$ V.

P3. Un β mayor mejora un margen de ruido y degrada el otro. ¿Cuál y por qué? ¿En qué aplicación preferiría un inversor deliberadamente descentrado?

Rpta: Un β mayor incrementa la conducción PMOS y desplaza la VTC a la derecha, aumentando V_{IL} (mejora NM_L) y reduciendo V_{IH} (degrada NM_H). Se emplea descentrado en receptores de línea ruidosos con referencia a tierra, comparadores con histéresis (Schmitt trigger) y en la lectura asimétrica de celdas SRAM.

P4. Repita el barrido con 45nm_LP.pm a $V_{DD} = 1.1$ V. ¿Cómo cambia la ganancia en la región de transición y qué parámetro de la tarjeta lo explica?

Rpta: La ganancia máxima de transición es mayor debido al óxido más grueso ($t_{oxe} = 1.80$ nm vs 1.25 nm en HP) y voltajes V_{TH0} superiores (≈ 0.62 V), lo que mitiga el efecto DIBL (ϵ_{ta0} , $pc1m$), reduciendo la conductancia g_{ds} y elevando la ganancia intrínseca $A_v = -g_m/(g_{ds,n} + g_{ds,p})$.



5. Actividad 2 — Respuesta transitoria: retardo y transición

Procedimiento

1. Use el netlist del Anexo A.2, configurado con el β óptimo de la Actividad 1 ($\beta = 2.5$, $W_p = 450$ nm, que centró la VTC en $V_M \approx 0.50$ V).
2. Mida t_{pHL} , t_{pLH} , t_r y t_f con las directivas `.meas` provistas.
3. Barra la carga con `.step param Cload if 8f if` y grafique t_p contra C_L . Ajuste una recta y determine su pendiente y su ordenada al origen (Figura 4).
4. Repita la medición para 90 nm, 45 nm HP y 22 nm HP, escalando V_{DD} y L según los parámetros nominales de cada nodo tecnológico y manteniendo $W_n = 4L$ y $C_L = 2$ fF (Tabla 3 y Figura 5).

Tabla 3: Resultados 2.1 — Retardo frente al nodo tecnológico ($\beta = 2.5$, $C_L = 2$ fF).

Nodo	V_{DD} (V)	t_{pHL} (ps)	t_{pLH} (ps)	t_p (ps)	t_r (ps)	t_f (ps)
90 nm	1.2	7.863	6.859	7.361	10.785	9.593
45 nm HP	1.0	7.881	4.929	6.405	7.686	11.814
22 nm HP	0.8	10.731	6.566	8.649	13.657	19.680

Figura 4: Ajuste lineal de retardo medio $t_p(C_L)$ en 45 nm HP ($\beta = 2.5$): $m = 1.846$ ps/fF, $t_0 = 2.708$ ps.Figura 5: Respuesta transitoria en 3 nodos ($C_L = 2$ fF): **Rojo** (22 nm HP, $t_p = 8.65$ ps), **Azul** (45 nm HP, $t_p = 6.41$ ps) y **Verde** (90 nm bulk, $t_p = 7.36$ ps).

Preguntas de la Actividad 2

P5. De la recta t_p contra C_L : ¿qué representa físicamente la pendiente y qué representa la ordenada al origen?

Rpta: La pendiente ($m = 1.846$ ps/fF) modela la resistencia equivalente media de conmutación ($m \approx 0.69 R_{eq}$), midiendo la degradación del retardo por fF externo. La ordenada al origen ($t_0 = 2.708$ ps) representa el retardo intrínseco descargado ($C_L = 0$), determinado por las capacitancias parásitas de unión (C_{db}) y solapamiento (C_{gd}).

P6. Con β correctamente ajustado, t_{pHL} y t_{pLH} deberían ser casi iguales. Si en su simulación difieren en más de 10 %, ¿qué corrección aplicaría y por qué el ajuste no resulta idéntico al que centró la VTC?

Rpta: Se corrige ajustando levemente el ancho PMOS para equilibrar las corrientes dinámicas de carga y descarga. El ajuste dinámico difiere del estático porque V_M evalúa un único punto DC en saturación ($V_{in} = V_{out}$), mientras que t_p integra la corriente no lineal a lo largo de toda la conmutación, donde los parásitos crecen con W_p .

P7. Entre 90 nm y 22 nm el retardo mejora aunque V_{DD} se reduce en un tercio. Identifique los dos mecanismos del escalamiento que compensan la caída de tensión.

Rpta: 1) La reducción de la longitud física de canal L y el adelgazamiento del óxido t_{ox} , que incrementan la densidad de corriente por unidad de ancho; y 2) La contracción geométrica del área de compuerta ($W \cdot L$), disminuyendo sensiblemente las capacitancias parásitas a conmutar en cada ciclo.

P8. Aumente el tiempo de subida del estímulo de 5 ps a 100 ps. ¿Qué componente de la ecuación (10) crece y por qué?

Rpta: Crece notablemente la potencia de cortocircuito ($I_{cc} V_{DD}$). Al ampliar el tiempo de subida de 5 ps a 100 ps, el circuito permanece durante una ventana temporal 20 veces mayor en el rango de conducción simultánea ($V_{Tn} < V_{in} < V_{DD} - |V_{Tp}|$), donde fluye corriente directa entre V_{DD} y tierra.

6. Actividad 3 — NAND2 y NOR2

Procedimiento

1. Implemente la NAND2 (Anexo A.3) y la NOR2 (Anexo A.4) con el criterio de **resistencia equivalente igual a la del inversor de referencia** ($\beta = 2.5$): en la rama serie los transistores duplican su ancho; en la rama paralela conservan el ancho del inversor.
2. Verifique la tabla de verdad de cada compuerta.
3. Mida t_{pHL} de la NAND2 en dos escenarios: (a) conmuta la entrada **interna** (la conectada al transistor cercano a tierra) con la otra fija en 1; (b) conmuta la entrada **externa** con la otra fija en 1.
4. Mida t_{pLH} de la NOR2 con las dos entradas conmutando simultáneamente y con solo una conmutando.
5. Compare NAND2 y NOR2 con idéntica carga: retardo promedio y ancho total de transistores (proxy de área).

Tabla 4: Resultados 3.1 — Comparación de compuertas con dimensionamiento $\beta = 2.5$ ($C_L = 2$ fF, 45 nm HP).

Compuerta	W_n /trans.	W_p /trans.	$\sum W$	t_{pHL} (ps)	t_{pLH} (ps)	t_p (ps)
Inversor (ref.)	180 nm	450 nm	630 nm	7.881	4.929	6.405
NAND2 (conm. ext.)	360 nm	450 nm	1620 nm	9.021	6.799	7.910
NAND2 (conm. int.)	360 nm	450 nm	1620 nm	10.221	7.724	8.972
NOR2 (una conm., B)	180 nm	900 nm	2160 nm	10.764	6.915	8.840
NOR2 (ambas conm.)	180 nm	900 nm	2160 nm	6.768	8.569	7.669

Preguntas de la Actividad 3

P9. Los dos escenarios del paso 3 dan retardos distintos. Explique el papel de la capacitancia del nodo intermedio y del efecto de cuerpo del transistor superior.

Rpta: En la conmutación externa (MN1), el nodo n_x ya estaba descargado a tierra por MN2, evacuando únicamente C_L ($t_{pHL} = 9.021$ ps). En la conmutación interna (MN2), n_x se encontraba precargado a $V_{DD} - V_{Tn}$ y debe descargarse a través de MN2; sumado al aumento de V_{Tn} por efecto de cuerpo en MN1 ($V_{SB} > 0$), el retardo se degrada a 10.221 ps.

P10. El esfuerzo lógico de la NAND2 es 4/3 y el de la NOR2 es 5/3. Verifique cualitativamente esa jerarquía con sus mediciones y explique por qué la industria construye bibliotecas dominadas por NAND.

Rpta: Las mediciones ratifican que NAND2 es más rápida en el peor caso individual ($t_p = 7.910$ ps vs 8.840 ps) y considerablemente más compacta ($\sum W = 1620$ nm vs 2160 nm, un 25 % menos de silicio). La industria prefiere NAND porque conecta en serie transistores NMOS (alta movilidad de electrones), duplicando un ancho pequeño ($W_n = 360$ nm); en contraste, la NOR apila PMOS (menor movilidad de huecos), forzando anchos masivos ($W_p = 900$ nm) que penalizan severamente el área de layout y la capacitancia parásita de entrada.

P11. ¿Qué le ocurriría al retardo de una NOR de cuatro entradas dimensionada con este mismo criterio? Estime el W_p requerido y comente su viabilidad.

Rpta: Requeriría 4 PMOS en serie con $W_p = 4 \times (\beta W_n) = 1800$ nm por dispositivo ($\sum W_p = 7.20$ μ m). El retardo de subida crecería de forma cuadrática ($\propto N^2$) por la acumulación de resistencia de canal y capacidades de difusión de los 3 nodos internos, resultando inviable en celdas estándar de alta velocidad.

P12. Si el sustrato del NMOS superior de la NAND2 se conectara a su fuente en lugar de a tierra, ¿qué cambiaría en el retardo? ¿Es esto físicamente realizable en un proceso *bulk*?

Rpta: Se anularía el efecto de cuerpo ($V_{SB} = 0$), preservando V_{Tn} en su mínimo nominal y reduciendo t_{pHL} . No es realizable en procesos estándar CMOS *bulk*, donde todos los NMOS

comparten el sustrato tipo p común polarizado a tierra; requeriría tecnologías de pozo triple (*triple-well*) o SOI.

7. Actividad 4 — Síntesis de una compuerta compleja

Diseñe, dimensione y verifique la compuerta que implementa

$$Y = \overline{(A \cdot B) + C} \quad (11)$$

Síntesis topológica y dimensionamiento

Se sintetiza la función inversora $Y = \overline{(A \cdot B) + C}$ mediante lógica CMOS estática. Conforme a la especificación de la netlist de referencia (Anexo A.5), se adopta el dimensionamiento basado en una razón entera $\beta = 2$ ($W_n = 180$ nm, $W_p = 360$ nm como referencia base), simplificando las razones geométricas en celdas multinivel y preservando el criterio de resistencia equivalente de conmutación ($R_{eq,n} = R_n$, $R_{eq,p} = R_p$):

- **Red Pull-Down (PDN):** A partir de la expresión no complementada $F = (A \cdot B) + C$, la operación AND se implementa con transistores NMOS en serie (MN_A, MN_B), colocados en paralelo con MN_C (operación OR). La rama serie duplica su resistencia equivalente ($2R_n$), requiriendo anchos duplicados $W_{nA} = W_{nB} = 2W_n = 360$ nm. La rama paralela conduce individualmente a tierra, manteniendo $W_{nC} = W_n = 180$ nm.
- **Red Pull-Up (PUN) dual:** Por dualidad topológica planar, la rama serie de la PDN se convierte en ramas paralelas ($MP_A \parallel MP_B$), las cuales se conectan en serie con MP_C . En el peor caso de conducción, solo conduce un PMOS de la rama paralela en serie con C ($R_{PUN} = R'_p + R'_p = 2R'_p$); para igualar $R_{PUN} = R_p$, cada PMOS debe duplicar su ancho: $W_{pA} = W_{pB} = W_{pC} = 2W_p = 720$ nm.




Figura 6: Diseño de la compuerta compleja $Y = \overline{(A \cdot B) + C}$: (a) Grafos de conducción duales; (b) Esquema circuital CMOS dimensionado ($W_n = 180$ nm, $\beta = 2$).

Verificación funcional y casos críticos de retardo

La Figura 7 presenta el barrido de las ocho combinaciones de entrada en 8 ns (111 a 000) según el Anexo A.5, verificando rigurosamente la tabla de verdad ($Y = 1$ únicamente para 100, 010 y 000):

- **Peor caso de carga (t_{pLH}):** Se mide directamente en la conmutación de $t = 3$ ns ($101 \rightarrow 100$, $t_{pLH} = 9.27$ ps) y $t = 5$ ns ($011 \rightarrow 010$, $t_{pLH} = 7.84$ ps), donde la corriente fluye por el camino más resistivo de la PUN (MP_C en serie con un solo PMOS en paralelo). El mejor caso ocurre en $t = 7$ ns ($001 \rightarrow 000$, $t_{pLH} = 6.47$ ps) al conducir ambas ramas PMOS en paralelo.
- **Análisis crítico de descarga (t_{pHL}):** El peor caso teórico ocurre al descargar exclusivamente por la rama serie $MN_A - MN_B$ ($ABC = 110$). Sin embargo, en el barrido continuo del Anexo A.5, el estado 110 es precedido por 111, por lo que la salida ya se encontraba en 0 V y no existe transición de bajada en esa ventana. Las caídas observadas en el barrido ($t = 4$ ns y $t = 6$ ns) descargan por el transistor rápido MN_C ($t_{pHL} = 6.92$ ps y 10.43 ps).

- **Medición dedicada del peor caso t_{pHL} :** Excitando un escalón hacia 110 desde un estado alto previo ($010 \rightarrow 110$ o $100 \rightarrow 110$), la descarga fluye exclusivamente por la rama serie, obteniéndose $t_{pHL} = 13.19$ ps para la entrada externa (A) y $t_{pHL} = 14.39$ ps para la interna (B , penalizada por la precarga del nodo intermedio n_{ab} y efecto cuerpo en MN_A).




Figura 7: Respuesta transitoria de la compuerta compleja $Y = \overline{(A \cdot B)} + C$ ante el barrido de las 8 combinaciones binarias (111 a 000) en 8 ns.

Preguntas de la Actividad 4

- P13.** ¿Cuántos transistores requiere su implementación? Compárela con la realización de la misma función mediante NAND2 más inversores en cascada, en número de transistores y en número de etapas.
- Rpta:** Requiere exactamente 6 transistores (3 NMOS y 3 PMOS) en una sola etapa inversora. Con celdas estándar exigiría una NAND2 (4T), inversor (2T), NOR2 (4T) e inversor (2T), totalizando 12 transistores y 3 retardos en cascada, duplicando el área y triplicando la latencia.
- P14.** La PUN de su diseño tiene una rama serie de dos PMOS. ¿Por qué esta función es más “amigable” para CMOS estático que su complemento $Y = \overline{(A + B) \cdot C}$?
- Rpta:** Porque la PUN de $Y = \overline{(A \cdot B)} + C$ limita a 2 los transistores PMOS en el camino crítico serie ($2W_p$). Su complemento requeriría mayor apilamiento PMOS en serie, forzando transistores mucho más anchos que degradarían fuertemente la capacitancia de entrada y el retardo de subida.
- P15.** Explique por qué la lógica CMOS estática siempre produce una función **inversora** y qué implica esto al mapear una expresión booleana arbitraria.
- Rpta:** Porque la red PDN conmuta a tierra con entradas en '1', mientras la PUN conduce a V_{DD} con entradas en '0', invirtiendo la polaridad lógica. Para implementar funciones no inversoras (AND u OR), es mandatorio sintetizar la función complementada y añadir un inversor CMOS a la salida.

8. Actividad 5 — Potencia: dinámica, cortocircuito y fuga

Procedimiento

1. **Fuga.** Con el netlist A.1 modificado para punto de operación: fije $V_{in} = 0$ y luego $V_{in} = V_{DD}$, y registre $|I(V1)|$ en cada caso. Repita con 45nm_HP.pm y 45nm_LP.pm.
2. **Dinámica.** En el netlist A.2, use `.meas tran Iavg AVG I(V1)` y calcule $P = |I_{avg}| V_{DD}$. Verifique contra $\alpha C_L V_{DD}^2 f$ y explique cualquier exceso.

Verificación de potencia dinámica del inversor ($C_L = 2$ fF, $f = 1$ GHz, $\alpha = 1$):

- **Proceso 45 nm HP ($V_{DD} = 1.0$ V, $I_{avg} = -2.754$ μ A):** $P_{teor} = \alpha C_L V_{DD}^2 f = (1)(2 \times 10^{-15} \text{ F})(1.0 \text{ V})^2(1 \times 10^9 \text{ Hz}) = 2.000 \mu\text{W}$; $P_{sim} = |I_{avg}| V_{DD} = |-2.754 \mu\text{A}|(1.0 \text{ V}) = 2.754 \mu\text{W} \Rightarrow \Delta P = 0.754 \mu\text{W}$.

- **Proceso 45 nm LP** ($V_{DD} = 1.1$ V, $I_{avg} = -3.097$ μ A): $P_{teor} = \alpha C_L V_{DD}^2 f = (1)(2 \times 10^{-15} \text{ F})(1.1 \text{ V})^2(1 \times 10^9 \text{ Hz}) = 2.420 \mu\text{W}$; $P_{sim} = |I_{avg}|V_{DD} = |-3.097 \mu\text{A}|(1.1 \text{ V}) = 3.407 \mu\text{W} \Rightarrow \Delta P = 0.987 \mu\text{W}$.

Discusión física de ΔP : El exceso ΔP evidencia que la potencia total real no es puramente capacitiva ideal; incorpora la componente de cortocircuito ($I_{cc}V_{DD}$) disipada durante la ventana en que ambos transistores conducen simultáneamente, sumada a la carga y descarga de las capacitancias parásitas intrínsecas de difusión (C_{db}, C_{sb}) de la celda.

3. **Oscilador de anillo.** Arme el anillo de cinco etapas del Anexo A.6. Mida el período de oscilación, deduzca $t_{pd} = T_{osc}/(2N)$ y compárelo con el t_p de la Actividad 2.
4. Repita el anillo con la tarjeta LP del mismo nodo y complete la tabla.

Tabla 5: Resultados 5.1 — Compromiso velocidad/fuga y potencia entre procesos HP y LP.

Proceso	V_{DD} (V)	$I_{leak}(0)$	$I_{leak}(V_{DD})$	f_{osc} (GHz)	t_{pd} (ps)	$P_{sim,inv}$ (μ W)	$P_{prom,RO}$ (μ W)
45 nm HP	1.0	4.131 nA	1.589 nA	21.94	4.56	2.754	177.2
45 nm LP	1.1	19.58 pA	10.98 pA	6.25	15.99	3.407	65.6




Figura 8: Oscilación transitoria estacionaria de las 5 fases en el oscilador de anillo (45 nm HP).

Preguntas de la Actividad 5

- P16.** Calcule las razones $I_{leak,HP}/I_{leak,LP}$ y $f_{osc,HP}/f_{osc,LP}$. Enuncie el compromiso de diseño en una frase y cite el parámetro de la tarjeta que lo origina.
Rpta: A partir de las corrientes promedio $I_{leak,HP} = (4.131 + 1.589)/2 = 2.860$ nA e $I_{leak,LP} = (19.58 + 10.98)/2 = 15.28$ pA, la razón es $I_{leak,HP}/I_{leak,LP} = 2.860 \text{ nA}/15.28 \text{ pA} \approx 187.2$ y $f_{osc,HP}/f_{osc,LP} = 21.94 \text{ GHz}/6.25 \text{ GHz} \approx 3.51$. Compromiso: “Acelerar la velocidad $3.5\times$ cuesta aumentar $\approx 187\times$ a $200\times$ el consumo estático”. El parámetro originario es el umbral V_{TH0} (fijado por `toxe` y `voff`), que modula exponencialmente la fuga subumbral.
- P17.** La corriente de fuga que midió no proviene de un solo mecanismo. Nombre al menos tres e indique con qué parámetros del modelo BSIM4 se controlan en las tarjetas PTM (revise `voff`, `nfactor`, `eta0`, `agidl`, `aigc`).
Rpta: 1) Fuga subumbral (I_{sub}): controlada por `voff`, `nfactor` y `eta0` (efecto DIBL); 2) Corriente de túnel de compuerta (I_{gate}): gobernada por `aigc` y espesor `toxe`; y 3) Fuga en unión drenador-sustrato inducida por compuerta (GIDL): modulada por `agidl` y `bgidl`.
- P18.** ¿Por qué el t_{pd} del oscilador de anillo suele ser menor que el t_p medido con un pulso ideal y carga fija?
Rpta: Porque en el inversor aislado de la Actividad 2 se impuso una carga externa de prueba ($C_L = 2$ fF) excitada con transiciones ideales; en el oscilador de anillo, cada etapa carga únicamente la compuerta de la etapa idéntica siguiente ($C_{in} \approx C_{ox}WL < 1$ fF), reduciendo la constante de tiempo RC.
- P19.** Para un procesador móvil que pasa 95% del tiempo en reposo, ¿qué proceso elegiría y qué técnica adicional aplicaría al 5% activo?

Rpta: Se elige el proceso 45 nm LP, cuya fuga 200 veces menor minimiza el drenaje estático de la batería durante el reposo. Durante el 5 % activo se aplica escalamiento dinámico de voltaje y frecuencia (DVFS) o polarización directa de sustrato (FBB) para recuperar transitoriamente la velocidad de cómputo.

P20. La potencia de cortocircuito no aparece explícitamente en sus mediciones porque está incluida en I_{avg} . Proponga un procedimiento de simulación que permita separarla de la componente de carga y descarga.

Rpta: Se simula con carga C_L obteniendo $E_{tot} = V_{DD} \int I(V_1) dt$. Luego se simula con $C_L = 0$ y se descuenta la energía de carga/descarga de parásitos internos C_{par} ; la energía remanente en la ventana de conducción mutua ($V_{Tn} < V_{in} < V_{DD} - |V_{Tp}|$) constituye la disipación neta de cortocircuito E_{cc} .

9. Conclusiones

- Saturación de velocidad y centrado estático:** En nodos sub-65 nm, el umbral lógico V_M y la simetría de la VTC están gobernados por la velocidad de saturación de portadores ($v_{sat,p}/v_{sat,n} \approx 0.88$) y no por la razón clásica de movilidades ($\mu_n/\mu_p \approx 2.70$), determinando que una relación $\beta \approx 2.50$ centra la conmutación y optimiza los márgenes de ruido ($NM_L \approx NM_H \approx 0.39$ V).
- Resistencia efectiva y retardo intrínseco:** El comportamiento lineal de $t_p(C_L)$ desacopla la resistencia equivalente de conmutación (pendiente $m \propto R_{eq}$) del retardo intrínseco descargado (ordenada $t_0 \propto R_{eq}C_{int}$). El escalamiento tecnológico preserva la velocidad de propagación pese a la reducción de V_{DD} debido a la contracción del área de compuerta y el adelgazamiento del óxido.
- Jerarquía de esfuerzo lógico (NAND vs. NOR):** La topología NAND2 demuestra superioridad en retardo de propagación y compacidad de silicio frente a NOR2, al concentrar el apilamiento serie en transistores NMOS de alta movilidad. En contraste, las compuertas NOR apilan dispositivos PMOS, requiriendo anchos que penalizan gravemente la capacitancia de entrada y tornan inviable su uso en abanicos de entrada elevados ($N \geq 4$).
- Densidad e integración en compuertas complejas AOI:** La síntesis directa de funciones compuestas como $Y = \overline{(A \cdot B) + C}$ en una única celda CMOS reduce a la mitad el conteo de transistores (6T frente a 12T) y consolida la transición a una sola etapa lógica, eliminando nodos intermedios de interconexión y maximizando el producto potencia-retardo-área.
- Compromiso fundamental entre desempeño y potencia estática:** La caracterización comparativa entre 45 nm HP y LP demuestra que un factor de $3.5\times$ en velocidad exige tolerar un incremento superior a $200\times$ en corriente de fuga estática. Este comportamiento sustenta la implementación de procesos LP en arquitecturas energéticamente restringidas y de procesos HP en bloques de procesamiento de alto rendimiento.

A. Netlists de referencia

Guarde cada bloque como archivo `.cir` y ejecútelo con **File** → **Open** en LTspice, o replique el circuito en esquemático.

A.1 Curva de transferencia del inversor

```
* === A1: VTC del inversor CMOS - PTM 45nm HP ===
.include 45nm_HP.pm

.param VDDv = 1.0
.param Lmin = 45n
.param Wn = 4*Lmin
.param beta = 2 ; Valor base de barrido (.step 1 a 3)

V1 vdd 0 {VDDv}
VIN in 0 0

MN out in 0 0 nmos L={Lmin} W={Wn}
MP out in vdd vdd pmos L={Lmin} W={beta*Wn}

.dc VIN 0 {VDDv} 2m
.step param beta 1 3 0.5
.meas dc VM FIND V(in) WHEN V(out)=V(in)
.end
```

Para el punto de fuga de la Actividad 5, reemplace las directivas de análisis por:

```
.dc VIN 0 {VDDv} {VDDv}
.meas dc Ileak_lo FIND I(V1) AT 0
.meas dc Ileak_hi FIND I(V1) AT {VDDv}
```

A.2 Respuesta transitoria

```
* === A2: Retardo de propagacion del inversor ===
.include 45nm_HP.pm

.param VDDv = 1.0
.param Lmin = 45n
.param Wn = 4*Lmin
.param beta = 2.5 ; beta optimo de referencia (Actividad 1)
.param Cload = 2f
```

```

V1 vdd 0 {VDDv}
VIN in 0 PULSE(0 {VDDv} 200p 5p 5p 500p 1n)

MN out in 0 0 nmos L={Lmin} W={Wn}
MP out in vdd vdd pmos L={Lmin} W={beta*Wn}
C1 out 0 {Cload}

.tran 0 2n 0 0.5p

.meas tran tPHL TRIG V(in) VAL={VDDv/2} TD=100p RISE=1 TARG V(out) VAL={
VDDv/2} FALL=1
.meas tran tPLH TRIG V(in) VAL={VDDv/2} TD=100p FALL=1 TARG V(out) VAL={
VDDv/2} RISE=1
.meas tran tf TRIG V(out) VAL={0.9*VDDv} FALL=1 TARG V(out) VAL={0.1*VDDv}
FALL=1
.meas tran tr TRIG V(out) VAL={0.1*VDDv} RISE=1 TARG V(out) VAL={0.9*VDDv}
RISE=1
.meas tran tp PARAM (tPHL+tpLH)/2
.meas tran Iavg AVG I(V1)

.step param Cload if 8f if
.end

```

A.3 NAND2 estática

```

* === A3: NAND2 CMOS estatica ===
.include 45nm_HP.pm

.param VDDv = 1.0
.param Lmin = 45n
.param Wn = 4*Lmin
.param beta = 2.5 ; beta optimo del inversor de referencia

V1 vdd 0 {VDDv}
VA a 0 PULSE(0 {VDDv} 200p 10p 10p 1n 2n)
VB b 0 {VDDv}

* --- PDN: NMOS serie ---
MN1 out a nx 0 nmos L={Lmin} W={2*Wn}
MN2 nx b 0 0 nmos L={Lmin} W={2*Wn}
* --- PUN: PMOS paralelo ---
MP1 out a vdd vdd pmos L={Lmin} W={beta*Wn}
MP2 out b vdd vdd pmos L={Lmin} W={beta*Wn}

C1 out 0 2f

.tran 0 4n 0 0.5p
.meas tran tPHL TRIG V(a) VAL={VDDv/2} RISE=1 TARG V(out) VAL={VDDv/2} FALL
=1
.meas tran tPLH TRIG V(a) VAL={VDDv/2} FALL=1 TARG V(out) VAL={VDDv/2} RISE
=1
.meas tran tp PARAM (tPHL+tpLH)/2
.end

```

A.4 NOR2 estática

```

* === A3: NOR2 CMOS estatica ===
.include 45nm_HP.pm
.param VDDv = 1.0
.param Lmin = 45n
.param Wn = 4*Lmin
.param beta = 2.5 ; beta optimo del inversor

V1 vdd 0 DC {VDDv}
VA a 0 0
VB b 0 PULSE(0 {VDDv} 1n 10p 10p 2n 4n)

* --- PUN: PMOS serie ---
MP1 y a vdd vdd pmos L={Lmin} W={2*beta*Wn}
MP2 out b y vdd pmos L={Lmin} W={2*beta*Wn}

* --- PDN: NMOS paralelo ---
MN1 out a 0 0 nmos L={Lmin} W={Wn}
MN2 out b 0 0 nmos L={Lmin} W={Wn}

C1 out 0 2f
.tran 0 8n 0 0.5p
.meas tran tPHL TRIG V(b) VAL={VDDv/2} RISE=1 TARG V(out) VAL={VDDv/2} FALL
=1
.meas tran tPLH TRIG V(b) VAL={VDDv/2} FALL=1 TARG V(out) VAL={VDDv/2} RISE

```

```

=1
.meas tran tp PARAM (tPHL+tpLH)/2
.end

```

A.5 Compuerta compleja AOI

```

* === A4: Compuerta Compleja AOI Y = -((A*B)+C) ===
.include 45nm_HP.pm
.param VDDv = 1.0
.param Lmin = 45n
.param Wn = 4*Lmin
.param beta = 2

V1 vdd 0 {VDDv}
VA a 0 PULSE(0 {VDDv} 0 10p 10p 4n 8n)
VB b 0 PULSE(0 {VDDv} 0 10p 10p 2n 4n)
VC c 0 PULSE(0 {VDDv} 0 10p 10p 1n 2n)

* --- PDN: (A serie B) paralelo C
MN_A out a nab 0 nmos L={Lmin} W={2*Wn}
MN_B nab b 0 0 nmos L={Lmin} W={2*Wn}
MN_C out c 0 0 nmos L={Lmin} W={Wn}

* --- PUN: C en serie con (A paralelo B)
MP_C vdd c pmid vdd pmos L={Lmin} W={2*beta*Wn}
MP_A pmid a out vdd pmos L={Lmin} W={2*beta*Wn}
MP_B pmid b out vdd pmos L={Lmin} W={2*beta*Wn}

C1 out 0 2f

.tran 0 8n 0 1p

* --- Mediciones robustas con ventana temporal (TD)
* Peor caso de subida (t = 3 ns: transicion ABC = 101 -> 100, flanco de
subida en out)
.meas tran tphl_worst TRIG V(c) VAL={VDDv/2} TD=2.5n FALL=1 TARG V(out) VAL
={VDDv/2} TD=2.5n RISE=1

* Descarga rapida por transistor C (t = 4 ns: transicion ABC = 100 -> 011,
flanco de bajada en out)
.meas tran tphl_c TRIG V(c) VAL={VDDv/2} TD=3.5n RISE=1 TARG V(out) VAL={
VDDv/2} TD=3.5n FALL=1

* Retardo promedio representativo
.meas tran tp_avg PARAM (tphl_worst + tphl_c)/2

.end

```

A.6 Oscilador de anillo de 5 etapas

```

* === A5: Oscilador de anillo, 5 etapas ===
.include 45nm_HP.pm

.param VDDv = 1.0
.param Lmin = 45n
.param Wn = 4*Lmin
.param beta = 2.5 ; Inversores calibrados de Actividad 1 y 2
.param N = 5

.subckt INV in out vdd
MN out in 0 0 nmos L={Lmin} W={Wn}
MP out in vdd vdd pmos L={Lmin} W={beta*Wn}
.ends

V1 vdd 0 {VDDv}
X1 n1 n2 vdd INV
X2 n2 n3 vdd INV
X3 n3 n4 vdd INV
X4 n4 n5 vdd INV
X5 n5 n1 vdd INV

.ic V(n1)=0
.tran 0 20n 5n 0.5p

.meas tran Tosc TRIG V(n1) VAL={VDDv/2} RISE=2 TARG V(n1) VAL={VDDv/2} RISE
=3
.meas tran fosc PARAM 1/Tosc
.meas tran tpd PARAM Tosc/(2*N)
.meas tran Iavg AVG I(V1)
.end

```